第十六章 DLL

對於一般的程式設計者來，VB所提供的基本功能可以說是綽綽有餘，但是如果想要設些特殊的功能，那麼勢必得要透過附加的Activex控制項……等其他的方式才能夠達成。

Activex控制項可以說是延伸VB功能的好選擇，但是並非你所需要解決的問題，都能夠找到一個適當的Activex控制項，反倒是透過其它不同方式，讓視窗這個系統本身或者其它的軟體來替你的程式進行服務的工作，這個時候，如何使用『動態連結程式庫(DLL，Dynamic-Link Libraries)』的功能，就成為一個很重要的技術。

什麼是『動態連結程式庫』呢?就平常VB的程式設計來說，並沒有使用到『程式庫(Libraries)』的程式設計概念，其實『程式庫』的概念相當地簡單，它就有如程式中的『模組』，在裡面存放著許多現成的程序(副程式)、函數，這些程式庫底下的程序具有『通用性』，可以在任何的程式中運用。其實VB本身能夠提供各項通用的函數、程序，所以它也是一個『程式庫』。

以程式設計的觀點來說，『程式庫的使用』可以分為兩種類型，分別是『靜態連結程式庫(Static-Link Libraries)』以及『動態連結程式庫』。所謂的『靜態連結程式庫』就是在應用程式碼之中，除了原本應用程式內部的程序之外，並且使用了外部程式庫的程序。在將程式碼編譯為應用程式的可執行檔時，編譯器會將所有的程序，無論它是程式庫的程序還是程式設計者所設計的程序，『連結(Link)』為同一個應用程式。因此一個使用了『靜態連結程式庫』的應用程式，它的內容除了程式設計者所設計的程式碼之外，還包含了所使用程式庫的程式碼，所以其檔案變得相當大。

若使用『動態連結』的方式，則在程式碼編譯為應用程式時，編譯器並不會將程式庫的程式碼加入應用程式之中，編譯器則為應用程式與程式庫之間建立一個能夠『互相溝通的管道』。在程式執行的過程中，應用程式如果要使用到程式庫的功能，系統才開始讓程式庫開始工作。因此如果在電腦中有超過一個以上的程式需要使用同樣一個『動態連結程式庫』的功能，那麼在系統中並不需要存有多個相同的動態連結程式庫，事實上，只需要『一個』動態連結程式庫即可，也就是說，相同的程式部分重複地被不同的程式使用，因此整體的程式運作所需要耗費的記憶體空間也就相對地減少。

DLL是Dynamic Link Libraries的縮寫，中文一般譯為動態連結程式庫。編譯階段作靜態的連結，這意謂著程式庫能夠獨立昇級而和應用程式無關，而且許多的應用程式可共享同一個DLL，實際上Windows作業系統就是由許多的DLL所組成，這些DLL含有許多Windows應用程式用來顯示視窗、繪圖、記憶體管理及周邊設備管理等等程序，這些程序救被稱為視窗應用軟體程式介面(Windows Application Programming Interface，簡稱API)，這和以前DOS時代的INT是相同的。

16-1　Window API

所謂的API，它的中文名稱為『應用軟體程式介面(Application Programming Interface)』，它的用途是提供一個界於『應用程式與系統環境』間溝通橋樑，使得應用程式能夠使用系統環境所提供的各項標準功能，建構出所需要的程式。

所以Windows API就是一個提供Windows作業系統與應用程式間的功能界面，透過這些API，應用程式才能夠建立視窗、指令按鈕…等物件，也才能夠讀取滑鼠的訊息、鍵盤的動作，在螢幕上繪製圖形…。

在使用Windows API之前，首先約略的介紹一個觀念，那就是handle(物件代碼)。在Windows的系統環境中，每一個程式中的控制項、每一個視窗、每一張圖片，這些『由系統所產生的物件』也都是以編號的方式來管理，這些『物件的編號』稱為handle(物件代碼)，只要根據某一個『物件的handle』，便可以針對它進行控制的工作。

handle在使用上，會按照不同的用途而分類，而這些不同用途的handle在Windows API的宣告時，也會使用不同的引數名稱。常見的handle種類如下:

	handle種類
	說明

	hWnd
	

	hDC
	

	hMenu
	

	hBitmap
	


在Windows API的引數命名中，通常如名稱前被冠上一小寫的h，即表示這一個handle數值，在VB的程式宣告中，需要以一個『傳值的長整數（long）來儲存hamdle值。一般VB的程式設計中，最常使用到的handle數值為『視窗』及『圖形裝置』的物件代碼(hWnd以及hDC)。

於\winapi資料夾，如下圖可找到windows API的常數及DLL程序宣告，使用者可於“專案”/“新增檔案”選單將win32api.txt加入。

範例16-1A:試使用DLL的Rectangle函數繪製矩形。

執行結果

設計步驟:

1. 表單物安排如下圖。
[image: image2.png]



2. 進入API檢視員，畫面如下圖左(進入API的方法有二，其一是按一下功能表的“增益集”/“API檢視員”；其二是按一下工作列的“開始”/“程式集”/VisualBasic6.0/“VB工具”/“API檢視員”。其次，若功能表的增益集下無“API檢視員”，請進入“增益功能管理員”〔於功能表的增益集選單內〕，請點選“VB API檢視員”，並勾選“載入/載出”與“啟動時載入”，如下圖右

3. 載入API文字檔，本例應載入Win32api，如下圖左(按一下功能表的“檔案”/“載入文字檔”，載入過程並提醒使用者是否轉為資料庫，若轉為資料庫則下次再載入時可提昇載入速度)

4. 搜尋欲使用的API函式宣告，本例欲使用Reclange，故鍵入rect，並按一下“加入”，畫面如下圖右

6.
將API函式宣告插入適當的表單或模組，本例欲插入表單的“一般宣告”。(按一下上圖右的“插入”，畫面如下圖左，按一下“確定”畫面如下圖右)

程式列印

Private Declare Function Rectangle Lib "gdi32" (ByVal hdc As Long, ByVal X1 As Long, ByVal Y1 As Long, ByVal X2 As Long, ByVal Y2 As Long) As Long

Private Sub cmdrect_Click()

    apireturn = Rectangle(frm.hdc, 50, 50, 300, 100)

End Sub
程式說明:

1. Rectangle函語法如下:

c=Rectangel(hdc,x1,y1,x2,y2)

(1)
如傳回值c值為False則繪圖失敗。

(2) hdc為圖形裝置如PictureBox或Form的handle值。

(3) 此函數於(X1,Y1)(X2,Y2)畫一直線。
(4) 若於32bits系統執行，則函數的大小寫必須一致才可執行。
相關知識:

1.
DLL的函數宣告可放在表單的般宣告區(General declarations)或模組(Moudle)，如於表單的一般宣告區則應加Private，若於模組則否(請看下範例)。

2. API的宣告與一般的程序宣告相同，其語法如下:

{Public/Private} Declare {Sub/Function} 程序名 Lib “DLL 檔名“（參數列）

(1)
程序若於表單(Form)中宣告，請加Private，此時此DLL只能於該表單呼叫。

(2)
程序若於模組(Module)中使用Private，則此程序只能於此模組使用。

(3)
VB若在16位元版本中執行，則函式名稱與大小寫無關，但若在32位元版本執行，則函式名稱大小寫必須相同才可呼叫。

(4)
常用的“DLL檔名”，如下表:

	API函數庫
	說明
	API函數庫
	說明

	kernel32.dll
	
	Shell32.all
	

	user32.dll
	
	winmm.dll
	

	gdi32.dll
	
	advapi32.dll
	

	comdgl32.dll
	
	Lz32.dll
	


3.
千萬不要把整個Win32api全部存檔至自已的專案中。(因為該檔案實在太大了)

4.
常用gdi32函數如下，讀者若有需要可自行研究。

(1) Ellipse 畫橢圓。

(2) MoveTo 畫直線。
(3) LineTo 畫直線。
(4) RoundRect 畫圓角四邊形。
(5) Arc畫弧線。
(6) Chard 畫弓形。
(7) Pic 畫扇形。
(8) Polyline 畫多個連續線段。
(9) Polygon 畫多邊形。
5.
使用DLL函數，只要參數錯誤，很容易當機，請於執行前先行存檔。

範例16-1B:試用DLL函數的GetVesion讀出系統Dos及Window版本。

Declare Function GetVersion Lib "kernel32" () As Long
Private Sub cmdstart_Click()

    Dim Version As Long

    Version = GetVersion()

    ver_string = Trim$(Hex$(Version))

    frm.Caption = ver_string

    ver_string = Right$("00000000" & ver_string, 8)

    txtdma.Text = Mid$(ver_string, 1, 2)

    txtdmi.Text = Mid$(ver_string, 3, 2)

    txtwmi.Text = Mid$(ver_string, 5, 2)

    txtwma.Text = Mid$(ver_string, 7, 2)

End Sub
16-2　QCARD.DLL

QCARD.DLL是由Stephen Marphy所創作的共享動態連結程式庫，主要是提供使用者撰寫樸克牌遊戲時一些圖片和函數，其使用方式請參考Qcard.hlp。本書範例磁片已將此一檔案載入使用者Windows\system子目錄下，讀者使用前請先心存感激Stephen Murphy一番。其次，若是32bits的使用者，請使用card32.dll。

範例16-2A:本範例示範

1. 從頭每次秀一張樸克牌。

2. 秀出任一花色的13張樸克牌。
3. 秀出6種背面圖案。
4. 洗牌後秀出52張牌。
5. 秀出可自定牌面長寬的牌。
執行結果:

1. 按一下“一張”按鈕，畫面如下圖。

2. 按一下“同花13”，畫面如下圖右:

3. 按一下“背面”按鈕，畫面如下圖。

4. 按一下“全部” 按鈕，畫面如下圖。

5. 按一下“可定大小”按鈕，畫面如下(本例於座標(30,40)處以寬度140，高度
80，顯示第5張牌)。

程式列印

1、card32.bas如下：

Option Explicit

#If Win32 Then

DefLng A-Z

#Else

DefInt A-Z

#End If

Public Const ordFaces = 0

Public Const ordBacks = 1

Public Const ordInvert = 2

Public Const ordCrossHatch = 53

Public Const ordPlaid = 54

Public Const ordWeave = 55

Public Const ordRobot = 56

Public Const ordRoses = 57

Public Const ordIvyBlack = 58

Public Const ordIvyBlue = 59

Public Const ordFishCyan = 60

Public Const ordFishBlue = 61

Public Const ordShell = 62

Public Const ordCastle = 63

Public Const ordBeach = 64

Public Const ordCardHand = 65

Public Const ordUnused = 66

Public Const ordX = 67

Public Const ordO = 68

Public Const ordClubs = 0

Public Const ordDiamonds = 13

Public Const ordHearts = 26

Public Const ordSpades = 39

#If Win32 Then

Declare Function cdtInit Lib "Cards32.Dll" ( _

    dx As Long, dy As Long) As Long

Declare Function cdtDrawExt Lib "Cards32.Dll" (ByVal hDC As Long, _

    ByVal x As Long, ByVal y As Long, ByVal dx As Long, ByVal dy As Long, _

    ByVal ordCard As Long, ByVal iDraw As Long, ByVal clr As Long) As Long

Declare Function cdtDraw Lib "Cards32.Dll" (ByVal hDC As Long, _

    ByVal x As Long, ByVal y As Long, _

    ByVal iCard As Long, ByVal iDraw As Long, ByVal clr As Long) As Long

Declare Function cdtAnimate Lib "Cards32.Dll" (ByVal hDC As Long, _

    ByVal iCardBack As Long, ByVal x As Long, ByVal y As Long, _

    ByVal iState As Long) As Long

Declare Function cdtTerm Lib "Cards32.Dll" () As Long

#Else

Declare Function cdtInit Lib "Cards.Dll" ( _

    dx As Integer, dy As Integer) As Integer

Declare Function cdtDrawExt Lib "Cards.Dll" (ByVal hDC As Integer, _

    ByVal x As Integer, ByVal y As Integer, _

    ByVal dx As Integer, ByVal dy As Integer, _

    ByVal iCard As Integer, ByVal iDraw As Integer, _

    ByVal clr As Long) As Integer

Declare Function cdtDraw Lib "Cards.Dll" (ByVal hDC As Integer, _

    ByVal x As Integer, ByVal y As Integer, _

    ByVal iCard As Integer, ByVal iDraw As Integer, _

    ByVal clr As Long) As Integer

Declare Function cdtAnimate Lib "Cards.Dll" ( _

    ByVal hDC As Integer, ByVal iCardBack As Integer, _

    ByVal x As Integer, ByVal y As Integer, _

    ByVal iState As Integer) As Integer

Declare Function cdtTerm Lib "Cards.Dll" Alias "CdtTerm" () As Integer

#End If
2、主程式如下：

Dim i, r, j, t, a, b As Integer

Private Sub cmd1_Click()

    frm.Cls

    Static c

    frm.Caption = "此張編號為 " & c

    Call cdtDraw(frm.hDC, 30, 40, c, 0, 0&)

        '於（30,40) 顯示第C張牌

    c = c + 1

    cmd1.Caption = "可連續按"

End Sub

Private Sub cmd13_Click()

    frm.Cls

   For i = 0 To 12

        Call cdtDraw(frm.hDC, 30 + 25 * i, 40, 4 * (0 + i), 0, 0)

   Next i

End Sub

Private Sub cmd52_Click()

    Dim a(52) As Integer

    Randomize

    frm.Cls

    '產生52張牌

    For i = 0 To 51

        a(i) = i

    Next i

    '洗牌

    For i = 0 To 51

        r = Int(Rnd() * 52)

        swap a(i), a(r)

    Next i

    '顯示

    For i = 0 To 3

        For j = 0 To 12

            Call cdtDraw(frm.hDC, 10 + 30 * j, 10 + 50 * i, a(13 * i + j), 0, 0)

        Next j

    Next i

End Sub

Private Sub cmdback_Click()

    frm.Cls

    For i = 53 To 68

        Call cdtDraw(frm.hDC, 30 + 25 * (i - 53), 40, i, 1, 0)

    Next i

        '第53至68 為背面花色

End Sub

Private Sub cmdend_Click()

    End

End Sub

Private Sub cmdscale_Click()

    frm.Cls

    Call cdtDrawExt(frm.hDC, 30, 40, 140, 80, 5, 0, 0)

        '於 （30,40) 以寬度140，高度80顯示第五張牌

End Sub

Private Sub Form_Load()

    Call cdtInit(30, 50)

        '初始化動作

End Sub

Public Sub swap(a, b)

    t = a: a = b: b = t

End Sub
範例16-2B:目前有一種遊戲，其工具為擁有數百張成對圖片。其玩法如下，首先將所有圖片整齊的蓋著排在桌上供若干人輪流取用，其規則為每人依序拿二張，若此二張牌相同，則該人可取走此二張牌且再取二張牌，若不同則展示後再蓋回原位置，直到牌面全部拿完為止。試以此方式設計一單人玩的紙牌遊戲。

執行結果:

1. 按一下“開始”鍵，畫面如下:

2. 使用滑鼠於任二張依序按一下如下圖(若此二張牌不同，電腦停留些許後
自動蓋回；若此二張牌相同電腦將自動移走此二張牌)。

3. 直到所有牌都取完為止。

Dim a(24), outed(3, 5), opened(3, 5), b(52)

Dim m, n, w, h As Integer

Dim second

Dim i1, j1, fi1, fj1

Dim card, cardf As Integer

Dim t1

Private Sub cmdend_Click()

    End

End Sub

Private Sub cmdstart_Click()

  t1 = Timer

  '顯示24張牌

  For i = 0 To n

        For j = 0 To m

            Call cdtDraw(frm.hDC, 10 + j * w, 10 + i * h, 64, 1, 0&)

        Next j

    Next i

End Sub

Private Sub Form_Load()

    Call cdtInit(20, 30)

    m = 5       '6行

    n = 3       '4列

    w = 80      '牌寬

    h = 100     '牌高

    Randomize

    ' 設定52張牌

    For i = 0 To 51

        b(i) = i

    Next i

   ' 洗牌

    For i = 0 To 51

        r = Rnd(52 * Rnd())

        swap b(i), b(r)

    Next i

    '取12張，每張2份

    For i = 0 To 11

        a(2 * i) = b(i)

        a(2 * i + 1) = b(i)

    Next i

    '洗牌

    For i = 0 To 23

        r = Int(24 * Rnd())

        swap a(r), a(i)

    Next i

    '預設24張狀態

    For i = 0 To n

        For j = 0 To m

            opened(i, j) = False  '未翻開

            outed(i, j) = False   '為取走

        Next j

    Next i

    second = False

End Sub

Public Sub swap(a, b)

    t = a: a = b: b = t

End Sub

Private Sub Form_MouseDown(Button As Integer, Shift As Integer, x As Single, y As Single)

    Static c

    If x < 10 Or y < 10 Then   '滑鼠所按的位置超過左邊界

        Exit Sub

    End If

    '將滑鼠座標轉為行列編號

    j1 = Int((x - 10) / w)

    i1 = Int((y - 10) / h)

    frm.Caption = x & " " & y & "  " & i1 & " " & j1

    If j1 > m Or i1 > n Then Exit Sub  '超過右邊界

    If outed(i1, j1) = True Then Exit Sub '該位置已取走

    If opened(i1, j1) = True Then Exit Sub '該位置已打開

    card = a(i1 * (m + 1) + j1) '二維陣列轉為一維陣列

    Call cdtDraw(frm.hDC, 10 + j1 * w, 10 + i1 * h, card, 0, 0)

            '顯示使用者所按的牌

    '停留數秒

    For i = 1 To 1000

        For j = 1 To 1000

        Next j

    Next i

    If second = True Then  ' 如果是第二張牌

        second = False

        If card = cardf Then  '如果2張相同

            '取走這2張牌

            Call cdtDraw(frm.hDC, 10 + fj1 * w, 10 + fi1 * h, cardf, 2, 0&)

            Call cdtDraw(frm.hDC, 10 + j1 * w, 10 + i1 * h, card, 2, 0&)

            outed(i1, j1) = True

            outed(fi1, fj1) = True

            c = c + 1

            If c = 12 Then

                Beep

                frm.Caption = (Timer - t1) \ 60 & "分" & (Timer - t1) Mod 60 & "秒"

                MsgBox "恭禧您已全部打開"

                Exit Sub

            End If

        Else

            '蓋回這2張牌

            Call cdtDraw(frm.hDC, 10 + fj1 * w, 10 + fi1 * h, 64, 1, 0)

            Call cdtDraw(frm.hDC, 10 + j1 * w, 10 + i1 * h, 64, 1, 0)

            opened(fi1, fj1) = False

            opened(i1, j1) = False

            cardf = -1

         End If

    Else

        '此為第一張牌

        second = True

        fi1 = i1

        fj1 = j1

        opened(fi1, fj1) = True

        cardf = card

    End If

End Sub

補充說明:

請先確認Windows/System資料夾下有Cards32.dll，此檔案本書範例磁片有付，請執行setup安裝程式。

範例16-2C:設有吃角子老虎，其規則如下:

1. 一對，(本例規定8以上)可得賭資1倍，否則莊家一定賠。

2. 二對，可得賭資2倍。

3. 三條，可得賭資3倍。

4. 順子，可得賭資4倍。

5. 同花，可得賭資6倍。

6. 葫蘆，可得賭資9倍。

7. 鐵枝，可得賭資25倍。

8. 同花順，可得賭資50倍。

9. 同花大順，可得賭資250倍。(本例規定10,J,Q,K,A且花色相同為同花大順)

10. 若不符以上之任一牌型，則沒入賭資。

執行結果:

1. 程式執行後，按一下“發牌”按鈕，畫面如下圖:

2. 可將不要的牌按一下，再按一下“換牌確定”鈕即可換牌，畫面如下圖:

[image: image13.png]= SRARE C EERL





程式列印

Dim a(52) As Integer

Dim same_color As Boolean  '同花

Dim straight As Boolean    '順子

Dim pair_num(2) As Integer '重複的點數

Dim pair_card(2) As Integer '重複的花色

Dim pa_num As Integer       '重複牌的組數

Dim flag(5) As Boolean  '

Dim cover(5) As Boolean    '

Dim big As Boolean

Dim point1

Dim b As String

Dim s As Integer

Dim biger As Boolean

Private Sub cmdend_Click()

    End

End Sub

Private Sub cmdok_Click()

    j = 5 '待換牌指標

    '將使用者蓋起來的牌換掉

    For i = 0 To 4

        If cover(i) = True Then

            a(i) = a(j)

            j = j + 1

        End If

    Next i

    '測試資料

    'a(0) = 0: a(1) = 36: a(2) = 40: a(3) = 48: a(4) = 44

    '        '同花大順

    'a(0) = 0: a(1) = 12: a(2) = 4: a(3) = 16: a(4) = 8

            '同花順

    'a(0) = 37: a(1) = 45: a(2) = 5: a(3) = 33: a(4) = 13

            '同花

    'a(0) = 6: a(1) = 33: a(2) = 7: a(3) = 34: a(4) = 35

            '葫蘆

    'a(0) = 8: a(1) = 33: a(2) = 7: a(3) = 34: a(4) = 35

            '三條

    'a(0) = 6: a(1) = 33: a(2) = 7: a(3) = 34: a(4) = 0

            'Two Pair

    'a(0) = 8: a(1) = 33: a(2) = 7: a(3) = 34: a(4) = 51

            '一對

    For i = 0 To 4

        Call cdtDraw(frm.hDC, 30 + i * 80, 50, a(i), 0, 0&)

    Next i

    judge_card

    lbl1.Caption = b   '牌型

    lbl2.Caption = s & " 倍"  '倍數

    lblmoney.Caption = lblmoney.Caption + s * txtmon.Text

    frm.Caption = "請輸入賭注，再按開始"

    cmdok.Enabled = False

    cmdstart.Enabled = True

    upd.Enabled = True

    txtmon.Enabled = True

End Sub

Private Sub cmdstart_Click()

    '產生52張牌

    For i = 0 To 51

        a(i) = i

    Next i

     '洗牌

    For i = 0 To 51

        r = Int(52 * Rnd())

        swap a(i), a(r)

    Next i

    '以前五張發給使用者，並顯示

    For i = 0 To 4

        Call cdtDraw(frm.hDC, 30 + i * 80, 50, a(i), 0, 0&)

    Next i

    judge_card  ' 判斷牌型

    lbl1.Caption = b

    lbl2.Caption = s

    frm.Caption = "請按一下欲換的牌，再按確定"

    cmdok.Enabled = True

    cmdstart.Enabled = False

    upd.Enabled = False

    txtmon.Enabled = False    

End Sub

Private Sub Form_Load()

    Call cdtInit(30, 50)

    frm.ScaleMode = 3

    Randomize

      s = -1

    same_color = True

    straight = False

    big = False

    pa_num = 0

    lbl1.Caption = ""

    For i = 0 To 1

        flag(i) = False

        cover(i) = False

    Next i

    cmdok.Enabled = False

    frm.Caption = "請按開始鍵"

End Sub

Public Sub judge_card()

    Dim staaight1 As Boolean

    Dim d(3) As String

    d(0) = "Club"

    d(1) = "Dimand"

    d(2) = "Heart"

    d(3) = "Spade"

    s = -1      '倍數

    same_color = True

    straight = False

    big = False

    pa_num = 0

    b = ""

    For i = 0 To 1

        pair_num(i) = 0

        pair_card(i) = 0

    Next i

    For i = 0 To 4

        flag(i) = False

        cover(i) = False

    Next i

    For i = 0 To 3

        If flag(i) Then GoTo aa

                '假如該牌已被前張牌構成牌型

                '則不用在比較

        point1 = a(i) \ 4 + 1 '點數

        color1 = a(i) Mod 4    '花色

        For j = i + 1 To 4

            point2 = a(j) \ 4 + 1

            color2 = a(j) Mod 4

            If color2 <> color1 Then same_color = False

                        '判斷同花

            If point2 = point1 Then

                pair_num(pa_num) = pair_num(pa_num) + 1

                                '該張牌重複次數

                flag(j) = True  '該牌已構成牌型

            End If

        Next j

        '假如第i張已構成牌型

        If pair_num(pa_num) > 0 Then

                            '第零對重複牌已找到

             pair_num(pa_num) = pair_num(pa_num) + 1

                            '加上被比較那一張（第i張）

             pair_card(pa_num) = point1

                            '記錄重複牌的點數

             pa_num = pa_num + 1

                            '再找下一組重複牌型

        End If

aa:

    Next i

    If pa_num = 1 Then      '只有一組牌有重複

        Select Case pair_num(0)

            Case 2

                b = pair_card(0) & " 一對"

                If pair_card(0) > 7 Or pair_card(0) = 1 Then

                    s = 1

                End If

                Exit Sub

            Case 3

                b = pair_card(0) & " 三條"

                s = 3

                Exit Sub

            Case 4

                b = pair_card(0) & " 鐵枝"

                s = 25

                Exit Sub

         End Select

         Exit Sub

    End If

    If pa_num = 2 Then  '兩種牌重複

        If pair_num(0) + pair_num(1) = 5 Then

            b = pair_card(0) & " " & pair_card(1) & " 葫蘆"

            s = 9

            Exit Sub

        Else

            b = pair_card(0) & "  " & pair_card(1) & " 雙對"

            s = 2

            Exit Sub

        End If

    End If

    biger = False

    judge_straight    '判斷順 Ace=0

    straight1 = straight

    biger = True

    judge_straight  '判斷大順 Ace=13

    straight = straight Or straight1

    If straight And same_color And big Then

        b = d(color1) & " 同花大順"

        s = 250

        Exit Sub

    End If

    If straight And same_color Then

        b = d(color1) & " 同花順"

        s = 50

        Exit Sub

    End If

    If straight Then

        b = " 順子"

        s = 4

        Exit Sub

    End If

    If same_color Then

        b = d(color1) & " 同花"

        s = 6

        Exit Sub

    End If

End Sub

Public Sub judge_straight()

    Static c(5) As Integer

    For i = 0 To 4

        c(i) = a(i) \ 4

        If c(i) = 0 And biger = True Then c(i) = 13

    Next i

    For i = 0 To 3

        For j = 0 To 3 - i

             If c(j) > c(j + 1) Then swap c(j), c(j + 1)

        Next j

    Next i

    For i = 0 To 3

        If c(i) + 1 <> c(i + 1) Then

             straight = False

             Exit Sub

        End If

    Next i

    straight = True

    If c(0) = 9 Then big = True

End Sub

Public Sub swap(a, b)

    t = a: a = b: b = t

End Sub

Private Sub Form_MouseDown(Button As Integer, Shift As Integer, x As Single, y As Single)

    If y < 50 Or y > 150 Then Exit Sub

    If x < 30 Or x > 420 Then Exit Sub

    nx = Int((x - 30) / 80)

    If cover(nx) = True Then

        Call cdtDraw(frm.hDC, 30 + nx * 80, 50, a(nx), 0, 0&)

        cover(nx) = False

    Else

        Call cdtDraw(frm.hDC, 30 + nx * 80, 50, 64, 1, 0&)

        cover(nx) = True

    End If

End Sub

Private Sub spi_SpinDown()

End Sub

Private Sub spi_SpinUp()

End Sub

Private Sub upd_DownClick()

    txtmon.Text = txtmon.Text - 5

End Sub

Private Sub upd_UpClick()

    txtmon.Text = CStr(CInt(txtmon.Text) + 5)

End Sub
相關知識:

1. 判斷是否同花的演算法如下:

同花=Ture

FOR I=0 TO 3

分別以前4張為基準點

FOR J=I+1 TO通4(逐次向右比較)

(1)
與基準牌比較，若有相同的點數，則累加該點數出現的次數，且該張牌作記號，待會不得當基準牌。
(2)
若興基準點花色不同，則同花=False。

NEXT J

若有重覆點數出現，則組數計數器加1 
NEXT I

2. 判斷順子的觀念如下:

(1)
將5張牌排序。

(2)
逐一比較排序後的序列是否前後兩兩點數差1。

習題

1. 試寫一程式可以印出指定花色的13張牌。

2. 試寫一程式可以印出指定點數的4張牌。

3. 試寫一程式摸擬樸克牌的“十點半”，其規則與功能如下

(1)
設定遊戲人數

(2)
按序每人發一張牌。

(3)
按序每人補牌，每人至多可補發4張牌，待遊戲者喊停為止，並主動為遊戲者計算點數(J、Q及K算半點)，其和為10點半者或5張牌的和小於10點半（五小）可直接得雙倍賭金。

(4)
每人均補完牌(含莊家)，其點數比莊家大者可得一倍賭金，否則沒入賭金。

4.
假設你手上沒有樸客牌，試設計一個程式可以讓你們玩“梭哈”“大老二”或“橋牌”等遊戲(可用4個表單個別顯示4個人的牌)。

5、請設計一程式，可讓人與電腦對完撿紅點。
